Pauli Pascal Triangle and other Noncommutative Expansions

by José Luis Gómez-Muñoz
http://homepage.cem.itesm.mx/lgomez/quantum/
jose.luis.gomez@itesm.mx
With contributions by Gianfranco Arroyo-Orrico and Sergio E. Martinez-Casas

Introduction

This is document shows algebraic expansions of binomials for operators with different commutation or anticommutation relationships.

Standard Pascal Triangle for Scalars

First we show the standard Pascal Triangle for commutative algebra:

```
Clear[a, b];
TraditionalForm[Grid[
  Table[{
    Expand[(a + b)^n]
   }, {n, 1, 7, 1}],
  Dividers → All]]
```

$a + b$
$a^2 + 2\, a\, b + b^2$
$a^3 + 3\, a^2\, b + 3\, a\, b^2 + b^3$
$a^4 + 4\, a^3\, b + 6\, a^2\, b^2 + 4\, a\, b^3 + b^4$
$a^5 + 5\, a^4\, b + 10\, a^3\, b^2 + 10\, a^2\, b^3 + 5\, a\, b^4 + b^5$
$a^6 + 6\, a^5\, b + 15\, a^4\, b^2 + 20\, a^3\, b^3 + 15\, a^2\, b^4 + 6\, a\, b^5 + b^6$
$a^7 + 7\, a^6\, b + 21\, a^5\, b^2 + 35\, a^4\, b^3 + 35\, a^3\, b^4 + 21\, a^2\, b^5 + 7\, a\, b^6 + b^7$

Coefficients of the standard Pascal Triangle:

```
Grid[
 Table[{
   expanded = Expand[(a + b)^n];
   Table[Coefficient[expanded, a^(n-j) b^j], {j, 0, n}]
  }, {n, 1, 7, 1}],
 Dividers → All]
```

{1, 1}
{1, 2, 1}
{1, 3, 3, 1}
{1, 4, 6, 4, 1}
{1, 5, 10, 10, 5, 1}
{1, 6, 15, 20, 15, 6, 1}
{1, 7, 21, 35, 35, 21, 7, 1}

In this Pascal triangle each new number is the addition of the two numbers above it.

Pauli-Pascal Triangle for Anticommuting Operators

We load the Quantum *Mathematica* add-on:

```
Needs["Quantum`Notation`"];
SetQuantumAliases[];
```

Press [ESC]anti[ESC] in order to show the anticommutator template (notice the + symbol in the subscript).
Here we assign the value of zero to the anticommutator of these two operators:

```
Clear[x, y];
SetQuantumObject[x, y];
⟦x, y⟧₊ = 0
```

```
0
```

Anticommutating operators generate the Pauli-Pascal Triangle. However, it does not look like a triangle, because the coefficients equal to zero are not shown. **Notice that "CommutatorExpand" is used instead of "Expand"**:

```
TraditionalForm[Grid[
  Table[{
    CommutatorExpand[(x + y)^n]
   }, {n, 1, 7, 1}],
  Dividers → All]]
```

$x + y$
$x^2 + y^2$
$x^2 y + x y^2 + x^3 + y^3$
$2 x^2 y^2 + x^4 + y^4$
$x^4 y + 2 x^3 y^2 + 2 x^2 y^3 + x y^4 + x^5 + y^5$
$3 x^4 y^2 + 3 x^2 y^4 + x^6 + y^6$
$x^6 y + 3 x^5 y^2 + 3 x^4 y^3 + 3 x^3 y^4 + 3 x^2 y^5 + x y^6 + x^7 + y^7$

Next all the coefficients are shown, including the zeros, and the threefold structure of the Pauli-Pascal Triangle can be easily seen:

```
Grid[
 Table[{
   expanded = CommutatorExpand[(x + y)^n];
   Table[Coefficient[expanded, x^(n-j) · y^j], {j, 0, n}]
  }, {n, 1, 9, 1}],
 Dividers → All]
```

{1, 1}
{1, 0, 1}
{1, 1, 1, 1}
{1, 0, 2, 0, 1}
{1, 1, 2, 2, 1, 1}
{1, 0, 3, 0, 3, 0, 1}
{1, 1, 3, 3, 3, 3, 1, 1}
{1, 0, 4, 0, 6, 0, 4, 0, 1}
{1, 1, 4, 4, 6, 6, 4, 4, 1, 1}

In the Pauli-Pascal triangle each new number is the addition of the two numbers above it, except when those two numbers and the number above them are all three equal, in that case the new number becomes zero.

Pascal Triangle for Commuting Operators

We load the Quantum *Mathematica* add-on:

```
Needs["Quantum`Notation`"];
SetQuantumAliases[];
```

Press [ESC]comm[ESC] in order to show the commutator template.
Here we assign the value of zero to the commutator of these two operators, they commute:

```
Clear[v, w];
SetQuantumObject[v, w];
[[v, w]]_ = 0
```

```
0
```

Commuting operators reproduce the Pascal triangle, however, the terms do not appear in the standard order. **Notice that "CommutatorExpand" is used instead of "Expand"**:

```
TraditionalForm[Grid[
  Table[{
    CommutatorExpand[(v + w)^n]
   }, {n, 1, 7, 1}],
  Dividers → All]]
```

$v + w$
$2\,v w + v^2 + w^2$
$3\,v^2 w + 3\,v w^2 + v^3 + w^3$
$4\,v^3 w + 6\,v^2 w^2 + 4\,v w^3 + v^4 + w^4$
$5\,v^4 w + 10\,v^3 w^2 + 10\,v^2 w^3 + 5\,v w^4 + v^5 + w^5$
$6\,v^5 w + 15\,v^4 w^2 + 20\,v^3 w^3 + 15\,v^2 w^4 + 6\,v w^5 + v^6 + w^6$
$7\,v^6 w + 21\,v^5 w^2 + 35\,v^4 w^3 + 35\,v^3 w^4 + 21\,v^2 w^5 + 7\,v w^6 + v^7 + w^7$

Coefficients of the Pascal Triangle for commuting operators

```
Grid[
 Table[{
   expanded = CommutatorExpand[(v + w)^n];
   Table[Coefficient[expanded, v^(n-j) · w^j], {j, 0, n}]
  }, {n, 1, 7, 1}],
 Dividers → All]
```

{1, 1}
{1, 2, 1}
{1, 3, 3, 1}
{1, 4, 6, 4, 1}
{1, 5, 10, 10, 5, 1}
{1, 6, 15, 20, 15, 6, 1}
{1, 7, 21, 35, 35, 21, 7, 1}

In this Pascal triangle each new number is the addition of the two numbers above it.

Expansions for Operators with a Commutator Equal to One

We load the Quantum *Mathematica* add-on:

```
Needs["Quantum`Notation`"];
SetQuantumAliases[];
```

Press [ESC]comm[ESC] in order to show the commutator template.
Here we assign the value of one to the commutator of these two operators:

```
Clear[q, p];
SetQuantumObject[q, p];
⟦q, p⟧_ = 1
```

```
1
```

Here are the expansions of integer powers of the sum of these two operators. **Notice that "CommutatorExpand" is used instead of "Expand"**:

```
TraditionalForm[Grid[
  Table[{
    CommutatorExpand[(p + q)^n]
   }, {n, 1, 7, 1}],
  Dividers → All]]
```

$p+q$
$2\,pq + p^2 + q^2 + 1$
$3\,p^2q + 3\,pq^2 + p^3 + 3\,p + q^3 + 3\,q$
$4\,p^3q + 6\,p^2q^2 + 4\,pq^3 + 12\,pq + p^4 + 6\,p^2 + q^4 + 6\,q^2 + 3$
$5\,p^4q + 10\,p^3q^2 + 10\,p^2q^3 + 30\,p^2q + 5\,pq^4 + 30\,pq^2 + p^5 + 10\,p^3 + 15\,p + q^5 + 10\,q^3 + 15\,q$
$6\,p^5q + 15\,p^4q^2 + 20\,p^3q^3 + 60\,p^3q + 15\,p^2q^4 + 90\,p^2q^2 + 6\,pq^5 + 60\,pq^3 + 90\,pq + p^6 + 15\,p^4 + 45\,p^2 + q^6 + 15\,q^4 + 45\,q^2 + 15$
$7\,p^6q + 21\,p^5q^2 + 35\,p^4q^3 + 105\,p^4q + 35\,p^3q^4 + 210\,p^3q^2 + 21\,p^2q^5 + 210\,p^2q^3 + 315\,p^2q + 7\,pq^6 + 105\,pq^4 + 315\,pq^2 + p^7 + 21\,p^5 + 105\,p^3 + 105\,p + q^7 + 21\,q^5 + 105\,q^3 + 105\,q$

This time the coefficients can be organized in a growing series of Pascal-like triangles:

```
TraditionalForm[Grid[
  Join[{Table[Column[{"Coefficients of ", p^(n-2 triangulo-j) . q^j},
      Alignment → Center], {triangulo, 0, 4}]},
   Transpose[
    Table[
     Table[
      expanded = CommutatorExpand[(p + q)^n];
      Table[
       If[p^(n-2 triangulo-j) . q^j =!= 1,
        Coefficient[expanded, p^(n-2 triangulo-j) . q^j],
        Select[expanded, And[FreeQ[#, p], FreeQ[#, q]] &]
       ],
       {j, 0, n - 2 triangulo}]
      , {n, 1, 9, 1}], {triangulo, 0, 4}]]],
  Dividers → All, ItemSize → Full]]
```

Coefficients of $p^{n-j}q^j$	Coefficients of $p^{-j+n-2}q^j$	Coefficients of $p^{-j+n-4}q^j$	
{1, 1}	{}	{}	
{1, 2, 1}	{1}	{}	
{1, 3, 3, 1}	{3, 3}	{}	
{1, 4, 6, 4, 1}	{6, 12, 6}	{3}	
{1, 5, 10, 10, 5, 1}	{10, 30, 30, 10}	{15, 15}	
{1, 6, 15, 20, 15, 6, 1}	{15, 60, 90, 60, 15}	{45, 90, 45}	
{1, 7, 21, 35, 35, 21, 7, 1}	{21, 105, 210, 210, 105, 21}	{105, 315, 315, 105}	
{1, 8, 28, 56, 70, 56, 28, 8, 1}	{28, 168, 420, 560, 420, 168, 28}	{210, 840, 1260, 840, 210}	
{1, 9, 36, 84, 126, 126, 84, 36, 9, 1}	{36, 252, 756, 1260, 1260, 756, 252, 36}	{378, 1890, 3780, 3780, 1890, 378}	{12

In each Pascal-like triangle each new number is the addition of the two numbers above it times a coefficient that depends on the triangle and the row.

Expansions for Operators with an Anticommutator Equal to One

We load the Quantum *Mathematica* add-on:

```
Needs["Quantum`Notation`"];
SetQuantumAliases[];
```

Press [ESC]anti[ESC] in order to show the anticommutator template (notice the + symbol in the subscript).
Here we assign the value of one to the anticommutator of these two operators:

```
Clear[r, s];
SetQuantumObject[r, s];
⟦r, s⟧₊ = 1
```

```
1
```

Here are the expansions of integer powers of the sum of these two operators:

```
TraditionalForm[Grid[
  Table[{
    CommutatorExpand[(r + s)^n]
   }, {n, 1, 7, 1}],
  Dividers → All]]
```

$r + s$
$r^2 + s^2 + 1$
$r^2 s + r s^2 + r^3 + r + s^3 + s$
$2 r^2 s^2 + r^4 + 2 r^2 + s^4 + 2 s^2 + 1$
$r^4 s + 2 r^3 s^2 + 2 r^2 s^3 + 2 r^2 s + r s^4 + 2 r s^2 + r^5 + 2 r^3 + r + s^5 + 2 s^3 + s$
$3 r^4 s^2 + 3 r^2 s^4 + 6 r^2 s^2 + r^6 + 3 r^4 + 3 r^2 + s^6 + 3 s^4 + 3 s^2 + 1$
$r^6 s + 3 r^5 s^2 + 3 r^4 s^3 + 3 r^4 s + 3 r^3 s^4 + 6 r^3 s^2 + 3 r^2 s^5 +$ $6 r^2 s^3 + 3 r^2 s + r s^6 + 3 r s^4 + 3 r s^2 + r^7 + 3 r^5 + 3 r^3 + r + s^7 + 3 s^5 + 3 s^3 + s$

This time the coefficients can be organized in a growing series of Pauli-Pascal-like triangles:

```
TraditionalForm[Grid[
  Join[{Table[Column[{"Coefficients of ", r^(n-2 triangulo-j) . s^j},
      Alignment → Center], {triangulo, 0, 4}]},
   Transpose[
    Table[
     Table[
      expanded = CommutatorExpand[(r + s)^n];
      Table[
       If[r^(n-2 triangulo-j) . s^j =!= 1,
        Coefficient[expanded, r^(n-2 triangulo-j) . s^j],
        Select[expanded, And[FreeQ[#, r], FreeQ[#, s]] &]
       ],
       {j, 0, n - 2 triangulo}],
      {n, 1, 9, 1}], {triangulo, 0, 4}]]],
  Dividers → All, ItemSize → Full]]
```

Coefficients of $r^{n-j}s^j$	Coefficients of $r^{-j+n-2}s^j$	Coefficients of $r^{-j+n-4}s^j$	Coefficients of $r^{-j+n-6}s^j$	Coefficients of $r^{-j+n-8}s^j$
{1, 1}	{}	{}	{}	{}
{1, 0, 1}	{1}	{}	{}	{}
{1, 1, 1, 1}	{1, 1}	{}	{}	{}
{1, 0, 2, 0, 1}	{2, 0, 2}	{1}	{}	{}
{1, 1, 2, 2, 1, 1}	{2, 2, 2, 2}	{1, 1}	{}	{}
{1, 0, 3, 0, 3, 0, 1}	{3, 0, 6, 0, 3}	{3, 0, 3}	{1}	{}
{1, 1, 3, 3, 3, 3, 1, 1}	{3, 3, 6, 6, 3, 3}	{3, 3, 3, 3}	{1, 1}	{}
{1, 0, 4, 0, 6, 0, 4, 0, 1}	{4, 0, 12, 0, 12, 0, 4}	{6, 0, 12, 0, 6}	{4, 0, 4}	{1}
{1, 1, 4, 4, 6, 6, 4, 4, 1, 1}	{4, 4, 12, 12, 12, 12, 4, 4}	{6, 6, 12, 12, 6, 6}	{4, 4, 4, 4}	{1, 1}

In the Pauli-Pascal triangle each new number is the addition of the two numbers above it times a coefficient that depends on the triangle and the row, except when those two numbers and the number above them are all three equal, in that case the new number becomes zero.

Expansions for Operators with a Commutator Equal to Another Operator

We load the Quantum *Mathematica* add-on:

```
Needs["Quantum`Notation`"];
SetQuantumAliases[];
```

Press [ESC]comm[ESC] in order to show the commutator template.
Here we assign a third operator as the commutator of two operators:

```
Clear[f, g, h];
SetQuantumObject[f, g, h];
⟦f, g⟧_ = h
```

```
h
```

Expansions evaluate the commutator of f with g to the third operator h, and generate unevaluated commutators of f with h and g with h.
Please notice the use of **Expand and CommutatorExpand** in this example:

```
TraditionalForm[Grid[
  Table[{
    Expand[CommutatorExpand[(f + g)^n]]
   }, {n, 1, 4, 1}],
  Dividers → All]]
```

$f+g$
$2fg+f^2+g^2-h$
$[f,h]+2[g,h]+3f^2g+3fg^2-3fh-3gh+f^3+g^3$
$3[f,h]g+g[f,h]+5f[g,h]+[f,h]f+3f[f,h]+3[g,h]g+$ $5g[g,h]+4f^3g+6f^2g^2-6f^2h+4fg^3-3fhg-9fgh-6g^2h+f^4+g^4+3h^2$

Here the three commutators are defined. Notice that defining the commutator ⟦h, f⟧₋ also defines the commutator ⟦f, h⟧₋
Please notice the use of **Expand and CommutatorExpand** in this example:

```
Clear[f, g, h];
SetQuantumObject[f, g, h];
⟦f, g⟧₋ = h;
⟦h, g⟧₋ = 0;
⟦h, f⟧₋ = 0;
TraditionalForm[Grid[
  Table[{
    Expand[CommutatorExpand[(f + g)^n]]
   }, {n, 1, 4, 1}],
  Dividers → All]]
```

$f+g$
$2fg+f^2+g^2-h$
$3f^2g+3fg^2-3fh-3gh+f^3+g^3$
$4f^3g+6f^2g^2-6f^2h+4fg^3-3fhg-9fgh-6g^2h+f^4+g^4+3h^2$

Expansions for Operators with Nonnested, Symbolic Commutators

We load the Quantum *Mathematica* add-on:

```
Needs["Quantum`Notation`"];
SetQuantumAliases[];
```

Here we define c and d as operators without a known commutator and without a known anticommutator

```
Clear[c, d];
SetQuantumObject[c, d];
```

The expansions are given in terms of the unkwon commutator [c,d].
Notice the use of **CommutatorExpand** and **Expand**:

```
TraditionalForm[Grid[
  Table[{
    Expand[CommutatorExpand[(c + d)^n]]
   }, {n, 1, 5, 1}],
  Dividers → All]]
```

$c + d$
$-[c, d] + 2\,cd + c^2 + d^2$
$-[c, d]c - 2\,[c, d]d - 2\,c[c, d] - d[c, d] + 3\,c^2 d + 3\,cd^2 + c^3 + d^3$
$-[c, d]c^2 - 3\,c^2[c, d] - 3\,[c, d]d^2 - d^2[c, d] - 2\,c[c, d]c - d[c, d]c - 3\,[c, d]cd - 5\,c[c, d]d - 2\,d[c, d]d - 3\,cd[c, d] + 2\,[c, d]^2 + 4\,c^3 d + 6\,c^2 d^2 + 4\,cd^3 + c^4 + d^4$
$-[c, d]c^3 - 4\,c^3[c, d] - 3\,c^2[c, d]c - 2\,c[c, d]c^2 - d[c, d]c^2 - 4\,[c, d]c^2 d - 9\,c^2[c, d]d - 6\,c^2 d[c, d] - 4\,[c, d]d^3 - d^3[c, d] - d^2[c, d]c - 2\,d^2[c, d]d - 6\,[c, d]cd^2 - 9\,c[c, d]d^2 - 3\,d[c, d]d^2 - 4\,cd^2[c, d] + 2\,[c, d]^2 c + 5\,[c, d]^2 d + 5\,c[c, d]^2 + 2\,d[c, d]^2 + 3\,[c, d]c[c, d] + 3\,[c, d]d[c, d] - 3\,cd[c, d]c - 7\,c[c, d]cd - 3\,d[c, d]cd - 7\,cd[c, d]d + 5\,c^4 d + 10\,c^3 d^2 + 10\,c^2 d^3 + 5\,cd^4 + c^5 + d^5$

Different (but equivalent) expansions are obtained with the option **ReverseOrdering->True** inside the CommutatorExpand command:

```
TraditionalForm[Grid[
  Table[{
    Expand[CommutatorExpand[(c + d)^n, ReverseOrdering → True]]
   }, {n, 1, 5, 1}],
  Dividers → All]]
```

$c + d$
$-[d, c] + 2\,dc + c^2 + d^2$
$-2\,[d, c]c - [d, c]d - c[d, c] - 2\,d[d, c] + 3\,dc^2 + 3\,d^2 c + c^3 + d^3$
$-3\,[d, c]c^2 - c^2[d, c] - [d, c]d^2 - 3\,d^2[d, c] - 3\,[d, c]dc - 2\,c[d, c]c - 5\,d[d, c]c - c[d, c]d - 2\,d[d, c]d - 3\,dc[d, c] + 2\,[d, c]^2 + 4\,dc^3 + 6\,d^2 c^2 + 4\,d^3 c + c^4 + d^4$
$-4\,[d, c]c^3 - c^3[d, c] - 2\,c^2[d, c]c - 6\,[d, c]dc^2 - 3\,c[d, c]c^2 - 9\,d[d, c]c^2 - c^2[d, c]d - 4\,dc^2[d, c] - [d, c]d^3 - 4\,d^3[d, c] - 4\,[d, c]d^2 c - 9\,d^2[d, c]c - 3\,d^2[d, c]d - c[d, c]d^2 - 2\,d[d, c]d^2 - 6\,d^2 c[d, c] + 5\,[d, c]^2 c + 2\,[d, c]^2 d + 2\,c[d, c]^2 + 5\,d[d, c]^2 + 3\,[d, c]c[d, c] + 3\,[d, c]d[d, c] - 3\,c[d, c]dc - 7\,d[d, c]dc - 7\,dc[d, c]c - 3\,dc[d, c]d + 5\,dc^4 + 10\,d^2 c^3 + 10\,d^3 c^2 + 5\,d^4 c + c^5 + d^5$

Expansions for Operators with Nested, Symbolic Commutators

We load the Quantum *Mathematica* add-on:

```
Needs["Quantum`Notation`"];
SetQuantumAliases[];
```

Here we define i and j as operators without a known commutator and without a known anticommutator

```
Clear[i, j];
SetQuantumObject[i, j];
```

The option NestedCommutators → True inside the command CommutatorExpand specifies that nested commutators can be generated

```
TraditionalForm[Grid[
  Table[{
    Expand[CommutatorExpand[(i + j)^n, NestedCommutators → True]]
   }, {n, 1, 5, 1}],
  Dividers → All]]
```

$i + j$
$-[i, j] + 2\,i\,j + i^2 + j^2$
$-3\,i[i, j] - 3\,j[i, j] + [i, [i, j]] + 2\,[j, [i, j]] + 3\,i^2\,j + 3\,i\,j^2 + i^3 + j^3$
$-6\,i^2[i, j] - 6\,j^2[i, j] + 2\,[i, [i, j]]j + 4\,i[i, [i, j]] + 2\,j[i, [i, j]] + 5\,i[j, [i, j]] + 8\,j[j, [i, j]] - 3\,i[i, j]j - 9\,i\,j[i, j] + 3\,[i, j]^2 - [i, [i, [i, j]]] - [j, [i, [i, j]]] - 3\,[j, [j, [i, j]]] + 4\,i^3\,j + 6\,i^2\,j^2 + 4\,i\,j^3 + i^4 + j^4$
$-10\,i^3[i, j] + 10\,i^2[i, [i, j]] + 9\,i^2[j, [i, j]] - 11\,i^2[i, j]j - 19\,i^2\,j[i, j] - 10\,j^3[i, j] + 5\,[i, [i, j]]j^2 + 3\,j^2[i, [i, j]] + 20\,j^2[j, [i, j]] - 7\,i[i, j]j^2 - 19\,i\,j^2[i, j] + 3\,[i, j]^2\,j - 3\,[i, [i, [i, j]]]j - 2\,[i, [i, j]][i, j] - 3\,[j, [i, j]][i, j] + 15\,i[i, j]^2 + 11\,j[i, j]^2 - 8\,[i, j][i, [i, j]] - 5\,i[i, [i, [i, j]]] - 2\,j[i, [i, [i, j]]] - 10\,[i, j][j, [i, j]] - 4\,i[j, [i, [i, j]]] - 3\,j[j, [i, [i, j]]] - 7\,i[j, [j, [i, j]]] - 15\,j[j, [j, [i, j]]] + 11\,i[i, [i, j]]j + 2\,j[i, [i, j]]j + i[j, [i, j]]j + [i, j]j[i, j] + 9\,i\,j[i, [i, j]] + 21\,i\,j[j, [i, j]] - 4\,i\,j[i, j]j + [i, [i, [i, [i, j]]]] + [j, [i, [i, [i, j]]]] + [j, [j, [i, [i, j]]]] + 4\,[j, [j, [j, [i, j]]]] + 4\,[[i, j], [i, [i, j]]] + 5\,[[i, j], [j, [i, j]]] + 5\,i^4\,j + 10\,i^3\,j^2 + 10\,i^2\,j^3 + 5\,i\,j^4 + i^5 + j^5$

Expansions for Operators that Commute with their Commutator

We load the Quantum *Mathematica* add-on:

```
Needs["Quantum`Notation`"];
SetQuantumAliases[];
```

Here a, b are defined as operators that commute with their commutator:

```
Clear[a, b];
SetQuantumObject[a, b];
⟦a, ⟦a, b⟧_-⟧_- = 0;
⟦b, ⟦a, b⟧_-⟧_- = 0;
```

The option NestedCommutators → True inside the command CommutatorExpand specifies that commutators of commutators must be generated. In this case, those nested commutators have a value of zero:

```
TraditionalForm[Grid[
  Table[{
    Expand[CommutatorExpand[(a + b)^n, NestedCommutators → True]]
   }, {n, 1, 5, 1}],
  Dividers → All]]
```

$a + b$
$-[a, b] + 2\,ab + a^2 + b^2$
$-3\,a[a, b] - 3\,b[a, b] + 3\,a^2b + 3\,ab^2 + a^3 + b^3$
$-6\,a^2[a, b] - 6\,b^2[a, b] - 3\,a[a, b]b - 9\,ab[a, b] + 3\,[a, b]^2 + 4\,a^3b + 6\,a^2b^2 + 4\,ab^3 + a^4 + b^4$
$-10\,a^3[a, b] - 11\,a^2[a, b]b - 19\,a^2b[a, b] - 10\,b^3[a, b] - 7\,a[a, b]b^2 - 19\,ab^2[a, b] + 3\,[a, b]^2b + 15\,a[a, b]^2 + 11\,b[a, b]^2 + [a, b]b[a, b] - 4\,ab[a, b]b + 5\,a^4b + 10\,a^3b^2 + 10\,a^2b^3 + 5\,ab^4 + a^5 + b^5$

Expansions for Operators with Nonnested, Symbolic Anticommutators

We load the Quantum *Mathematica* add-on:

```
Needs["Quantum`Notation`"];
SetQuantumAliases[];
```

Here we define k and m as operators without a known commutator and without a known anticommutator

```
Clear[k, m];
SetQuantumObject[k, m];
```

The option Anticommutators → True inside the command CommutatorExpand specifies that anticommutators should be used in the expansion

```
TraditionalForm[Grid[
  Table[{
    Expand[CommutatorExpand[(k + m)^n, Anticommutators → True]]
   }, {n, 1, 5, 1}],
  Dividers → All]]
```

$k + m$
$\{k, m\} + k^2 + m^2$
$\{k, m\}k + m\{k, m\} + k^2m + km^2 + k^3 + m^3$
$\{k, m\}k^2 + k^2\{k, m\} + \{k, m\}m^2 + m^2\{k, m\} + m\{k, m\}k + \{k, m\}km - k\{k, m\}m + km\{k, m\} + 2\,k^2m^2 + k^4 + m^4$
$\{k, m\}k^3 + k^2\{k, m\}k + m\{k, m\}k^2 - k^2\{k, m\}m + 2\,k^2m\{k, m\} + m^3\{k, m\} + m^2\{k, m\}k + 2\,\{k, m\}km^2 - k\{k, m\}m^2 + m\{k, m\}m^2 - \{k, m\}^2m - k\{k, m\}^2 + \{k, m\}k\{k, m\} + \{k, m\}m\{k, m\} + km\{k, m\}k + k\{k, m\}km + m\{k, m\}km + km\{k, m\}m + k^4m + 2\,k^3m^2 + 2\,k^2m^3 + km^4 + k^5 + m^5$

Expansions for Operators with Nested, Symbolic Anticommutators

We load the Quantum *Mathematica* add-on:

```
Needs["Quantum`Notation`"];
SetQuantumAliases[];
```

Here we define t and u as operators without a known commutator and without a known anticommutator

```
Clear[t, u];
SetQuantumObject[t, u];
```

The options Anticommutators → True and NestedCommutators → True inside the command CommutatorExpand specifies that nested anticommutators should be used in the expansion

```
TraditionalForm[Grid[
  Table[{
    Expand[Expand[
      CommutatorExpand[(t + u)^n, Anticommutators → True, NestedCommutators → True]]]
   }, {n, 1, 5, 1}],
  Dividers → All]]
```

$t+u$
$\{t, u\} + t^2 + u^2$
$-t\{t, u\} + u\{t, u\} + \{t, \{t, u\}\} + t^2 u + t u^2 + t^3 + u^3$
$2 t^2\{t, u\} + 2 u^2\{t, u\} - \{t, \{t, u\}\}u - 2 t\{t, \{t, u\}\} - u\{t, \{t, u\}\} - 4 t\{u, \{t, u\}\} - 2 u\{u, \{t, u\}\} + 2 t\{t, u\}u + 6 tu\{t, u\} - \{t, u\}^2 + \{t, \{t, \{t, u\}\}\} + 2 \{u, \{t, \{t, u\}\}\} + \{u, \{u, \{t, u\}\}\} + 2 t^2u^2 + t^4 + u^4$
$-2 t^3\{t, u\} + 4 t^2\{t, \{t, u\}\} + 2 t^2\{u, \{t, u\}\} - 4 t^2\{t, u\}u + 2 t^2u\{t, u\} + 2 u^3\{t, u\} - 2 \{t, \{t, u\}\}u^2 + 4 u^2\{t, \{t, u\}\} - 2 u^2\{u, \{t, u\}\} + 4 t\{t, u\}u^2 - 14 tu^2\{t, u\} + 2 \{t, u\}^2u - 2 \{t, \{t, \{t, u\}\}\}u - \{u, \{t, \{t, u\}\}\}u - \{t, \{t, u\}\}\{t, u\} + 3 \{u, \{t, u\}\}\{t, u\} - 5 t\{t, u\}^2 - 5 u\{t, u\}^2 - 5 \{t, u\}\{t, \{t, u\}\} - 3 t\{t, \{t, \{t, u\}\}\} - u\{t, \{t, \{t, u\}\}\} - 5 \{t, u\}\{u, \{t, u\}\} - 4 t\{u, \{t, \{t, u\}\}\} - 8 u\{u, \{t, \{t, u\}\}\} - 9 t\{u, \{u, \{t, u\}\}\} + u\{u, \{u, \{t, u\}\}\} + 5 t\{t, \{t, u\}\}u + 2 t\{u, \{t, u\}\}u + 2 \{t, u\}u\{t, u\} + 7 tu\{t, \{t, u\}\} + 22 tu\{u, \{t, u\}\} - 4 tu\{t, u\}u + \{t, \{t, \{t, \{t, u\}\}\}\} + 2 \{u, \{t, \{t, \{t, u\}\}\}\} + 5 \{u, \{u, \{t, \{t, u\}\}\}\} + 3 \{\{t, u\}, \{t, \{t, u\}\}\} + \{\{t, u\}, \{u, \{t, u\}\}\} + t^4u + 2 t^3u^2 + 2 t^2u^3 + tu^4 + t^5 + u^5$

by José Luis Gómez-Muñoz
http://homepage.cem.itesm.mx/lgomez/quantum/
jose.luis.gomez@itesm.mx